
Change 1
	Function name
	f_TC_8_1_2_1_1_NR5GC()

	Reason for change
	UE test loop mode should be deactivated before postamble

	Summary of change
	Add close and deactivation of Ue test loop mode 

	TTCN module
	RRC_Reconfiguration_NR5GC

	MCC160 Comment
	Accepted. Added further improvements, see MCC160 implementation below.


Before Change
function f_TC_8_1_2_1_1_NR5GC() runs on NR5GC_PTC

  {// RRC reconfiguration / DRB / SRB / Establishment / Modification / Release / Success

   // @sic R5-197057 sic@

    f_NR5GC_Init(NR_1);

    // Create and configure cell

    f_NR_CellConfig_Def(nr_Cell1);

    // Preamble: If pc_IP_Ping is set to TRUE then, the UE is in 5GS state 1N-A, PDU SESSION ACTIVE according to TS 38.508-1, clause 4.4A.2 Table 4.4A.2-1

    // else, the UE is in 5GS state 1N-A and Test Loop Function (On) with UE test loop mode B on NR Cell 1 according to 38.508-1, clause 4.4A.2 Table 4.4A.2-3.

    if (pc_IP_Ping) {

       f_NR5GC_Preamble (nr_Cell1, STATE_IDLE_1A);

    }

    else {

       f_NR5GC_Preamble(nr_Cell1, STATE_IDLE_1A, TEST_LOOPModeB_ON);

    }

    f_NR_TestBody_Set(true);

    f_TC_8_1_2_1_1_NR5GC_TestBody();

    f_NR_TestBody_Set(false);

    // Switch/Power off UE

    f_NR_Postamble(nr_Cell1, STATE_CONNECTED_3A);

  } //end of f_TC_8_1_2_1_1_NR5GC
After Change
   function f_TC_8_1_2_1_1_NR5GC() runs on NR5GC_PTC

  {// RRC reconfiguration / DRB / SRB / Establishment / Modification / Release / Success

   // @sic R5-197057 sic@

    f_NR5GC_Init(NR_1);

    // Create and configure cell

    f_NR_CellConfig_Def(nr_Cell1);

    // Preamble: If pc_IP_Ping is set to TRUE then, the UE is in 5GS state 1N-A, PDU SESSION ACTIVE according to TS 38.508-1, clause 4.4A.2 Table 4.4A.2-1

    // else, the UE is in 5GS state 1N-A and Test Loop Function (On) with UE test loop mode B on NR Cell 1 according to 38.508-1, clause 4.4A.2 Table 4.4A.2-3.

    if (pc_IP_Ping) {

       f_NR5GC_Preamble (nr_Cell1, STATE_IDLE_1A);

    }

    else {

       f_NR5GC_Preamble(nr_Cell1, STATE_IDLE_1A, TEST_LOOPModeB_ON);

    }

    f_NR_TestBody_Set(true);

    f_TC_8_1_2_1_1_NR5GC_TestBody();

    f_NR_TestBody_Set(false);
   f_NR5GC_OpenUE_TestLoopMode_Deactivate_TestMode(nr_Cell1);
    // Switch/Power off UE

    f_NR_Postamble(nr_Cell1, STATE_CONNECTED_3A);

  } //end of f_TC_8_1_2_1_1_NR5GC
MCC160 Implementation
   function f_TC_8_1_2_1_1_NR5GC() runs on NR5GC_PTC

  {// RRC reconfiguration / DRB / SRB / Establishment / Modification / Release / Success

   // @sic R5-197057 sic@

    f_NR5GC_Init(NR_1);

    // Create and configure cell

    f_NR_CellConfig_Def(nr_Cell1);

    // Preamble: If pc_IP_Ping is set to TRUE then, the UE is in 5GS state 1N-A, PDU SESSION ACTIVE according to TS 38.508-1, clause 4.4A.2 Table 4.4A.2-1

    // else, the UE is in 5GS state 1N-A and Test Loop Function (On) with UE test loop mode B on NR Cell 1 according to 38.508-1, clause 4.4A.2 Table 4.4A.2-3.

    f_NR5GC_Preamble(nr_Cell1, STATE_IDLE_1A, PING_Or_TEST_LOOPModeB_ON);
    f_NR_TestBody_Set(true);

    f_TC_8_1_2_1_1_NR5GC_TestBody();

    f_NR_TestBody_Set(false);
   f_NR5GC_OpenUE_TestLoopMode_Deactivate_TestMode(nr_Cell1);
    // Switch/Power off UE

    f_NR_Postamble(nr_Cell1, STATE_CONNECTED_3A);

  } //end of f_TC_8_1_2_1_1_NR5GC
Change 2
	Function name
	f_TC_8_1_2_1_1_NR5GC_TestBody()

	Reason for change
	At step 14 and 15, the RRCReconfigurationComplete on SRB1 and PDU SESSION RELEASE COMPLETE on SRB2 can be received in any order 

	Summary of change
	Pas new parameter to f_NR_RRCReconfigNewDRBWithSSConfig() so that RRCReconfigurationComplete is received outside the function, and add interleave block to handle both messages.

	TTCN module
	RRC_Reconfiguration_NR5GC

	MCC160 Comment
	Accepted but implemented differently in order to set the pass verdict of Step14 out of the interleave to avoid compiler issues. See MCC160 implementation below.


Before Change
 function f_TC_8_1_2_1_1_NR5GC_TestBody() runs on NR5GC_PTC

   {

 ………………………
    // @siclog "Step 12A" siclog@

    // @sic R5-197057 sic@  Check: Does the test result of generic test procedure in TS 38.508-1 subclause 4.9.1

    // indicate that the UE is capable of exchanging IP data on DRB#1 established in Step 11?

    f_NR5GC_CheckDataPath(nr_Cell1, PDN_2, dedicatedBearer, v_DRBInfo.DRB_Id, "Step 12A");

    // @siclog "Steps 13-14" siclog@

    // The SS transmits an RRCReconfiguration message with a drb-ToReleaseList and PDU SESSION RELEASE COMMAND

    // Check: Does the UE transmit an RRCReconfigurationComplete message?

    f_NR_RRCReconfigNewDRBWithSSConfig(nr_Cell1,

                                       v_RadioBearerConfig,

                                       {v_DRBInfo.SS_Config}, // re-establish DRB in SS with exactly the same config as earlier

                                       v_CellGroupConfig,

                                       { cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_PiggyBackedMsgToSend) },

                                       tsc_NR_RbId_SRB1); // @sic R5s190388 sic@

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 14");

    // @siclog "Step 15 - Void" siclog@

    // @siclog "Step 16" siclog@

    // The UE transmits an ULInformationTransfer message and a UL NAS TRASPORT containing PDU SESSION RELEASE COMPLETE.

    SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell1,

                                      tsc_NR_RbId_SRB2,

                                      cr_NG_NAS_IndWithPiggybacking((tsc_SHT_IntegrityProtected_Ciphered,tsc_SHT_IntegrityProtected)))) -> value v_ReceivedAsp; // @sic R5s190388 sic@

    if (not f_Check_NG_PDUSessionReleaseComplete (v_PDUSessionInfoList, v_ReceivedAsp.Signalling.Nas[0].Pdu, cr_NG_PDU_SessionId(-, int2oct(v_DRBInfo.PDUSessionId, 1)))) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 15");

    }

   }//end of f_TC_8_1_2_1_1_NR5GC_TestBody

After Change
   function f_TC_8_1_2_1_1_NR5GC_TestBody() runs on NR5GC_PTC

   {

 ………………………
    // @siclog "Step 12A" siclog@

    // @sic R5-197057 sic@  Check: Does the test result of generic test procedure in TS 38.508-1 subclause 4.9.1

    // indicate that the UE is capable of exchanging IP data on DRB#1 established in Step 11?

    f_NR5GC_CheckDataPath(nr_Cell1, PDN_2, dedicatedBearer, v_DRBInfo.DRB_Id, "Step 12A");

    // @siclog "Steps 13-14" siclog@

    // The SS transmits an RRCReconfiguration message with a drb-ToReleaseList and PDU SESSION RELEASE COMMAND

    // Check: Does the UE transmit an RRCReconfigurationComplete message?

    f_NR_RRCReconfigNewDRBWithSSConfig(nr_Cell1,

                                       v_RadioBearerConfig,

                                       {v_DRBInfo.SS_Config}, // re-establish DRB in SS with exactly the same config as earlier

                                       v_CellGroupConfig,

                                       { cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_PiggyBackedMsgToSend) },

                                       tsc_NR_RbId_SRB1,false); // @sic R5s190388 sic@

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 14");

    // @siclog "Step 15 - Void" siclog@

    // @siclog "Step 16" siclog@

interleave

{

                  [] SRB.receive(car_NR_SRB_RrcPdu_IND(nr_Cell1, tsc_NR_RbId_SRB1, cr_38508_RRCReconfigurationComplete))                

{                      f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 14");                    } 

          // The UE transmits an ULInformationTransfer message and a UL NAS TRASPORT containing PDU SESSION RELEASE COMPLETE.

                 []SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell1,

                                      tsc_NR_RbId_SRB2,

  cr_NG_NAS_IndWithPiggybacking((tsc_SHT_IntegrityProtected_Ciphered,tsc_SHT_IntegrityProtected)))) ->value v_ReceivedAsp{}

}

    if (not f_Check_NG_PDUSessionReleaseComplete (v_PDUSessionInfoList, v_ReceivedAsp.Signalling.Nas[0].Pdu, cr_NG_PDU_SessionId(-, int2oct(v_DRBInfo.PDUSessionId, 1)))) 

{       f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 15");     }     }
   }//end of f_TC_8_1_2_1_1_NR5GC_TestBody
MCC160 Implementation
   function f_TC_8_1_2_1_1_NR5GC_TestBody() runs on NR5GC_PTC

   {

 ………………………
        // @siclog "Step 12A" siclog@

    // @sic R5-197057 sic@  Check: Does the test result of generic test procedure in TS 38.508-1 subclause 4.9.1

    // indicate that the UE is capable of exchanging IP data on DRB#1 established in Step 11?

    f_NR5GC_CheckDataPath(nr_Cell1, PDN_2, dedicatedBearer, v_DRBInfo.DRB_Id, "Step 12A");

    // @siclog "Steps 13-14" siclog@

    // The SS transmits an RRCReconfiguration message with a drb-ToReleaseList and PDU SESSION RELEASE COMMAND

    // Check: Does the UE transmit an RRCReconfigurationComplete message?

    f_NR_RRCReconfigNewDRBWithSSConfig(nr_Cell1,

                                       v_RadioBearerConfig,

                                       {v_DRBInfo.SS_Config}, // re-establish DRB in SS with exactly the same config as earlier

                                       v_CellGroupConfig,

                                       { cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_PiggyBackedMsgToSend) },

                                       tsc_NR_RbId_SRB1, -, false); // @sic R5s190388 sic@

    // @siclog "Step 15 - Void" siclog@

    // @siclog "Step 16" siclog@

    // The UE transmits an ULInformationTransfer message and a UL NAS TRASPORT containing PDU SESSION RELEASE COMPLETE.

    interleave

    {

      [] SRB.receive(car_NR_SRB_RrcPdu_IND(nr_Cell1, tsc_NR_RbId_SRB1, cr_38508_RRCReconfigurationComplete))                

      {}

      // The UE transmits an ULInformationTransfer message and a UL NAS TRASPORT containing PDU SESSION RELEASE COMPLETE.

      []SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell1,

                                          tsc_NR_RbId_SRB2,

                                          cr_NG_NAS_IndWithPiggybacking((tsc_SHT_IntegrityProtected_Ciphered,tsc_SHT_IntegrityProtected)))) -> value v_ReceivedAsp{}

    }
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 14");
    if (not f_Check_NG_PDUSessionReleaseComplete (v_PDUSessionInfoList, v_ReceivedAsp.Signalling.Nas[0].Pdu, cr_NG_PDU_SessionId(-, int2oct(v_DRBInfo.PDUSessionId, 1)))) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 15");

    }

   }//end of f_TC_8_1_2_1_1_NR5GC_TestBody
Change 3
	Function name
	f_NR_RRCReconfigNewDRBWithSSConfig()

	Reason for change
	At step 14 and 15, the RRCReconfigurationComplete on SRB1 and PDU SESSION RELEASE COMPLETE on SRB2 can be received in any order

	Summary of change
	Add new flag to control handling of RRCRecponfiugrationComplete, and pass to    f_NR_SendRRCReconfiguration()

	TTCN module
	NR5GC_RRCSteps

	MCC160 Comment
	Accepted


Before Change
 function f_NR_RRCReconfigNewDRBWithSSConfig(NR_CellId_Type p_CellId,

                                              template (value) RadioBearerConfig p_RadioBearerConfig,

                                              template (value) NR_RadioBearerList_Type p_SS_Config,

                                              template (value) CellGroupConfig p_CellGroupConfig,

                                              template (omit) NG_NAS_MSG_RequestList_Type p_NAS_Msg,

                                              NR_RadioBearerId_Type p_SRB := tsc_NR_RbId_SRB1,

                                              template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now) runs on NR5GC_PTC

  {

    var NR_SecurityParams_Type v_SecurityParams := f_NR_Security_Get();

    var template (value) RRCReconfiguration_v1530_IEs v_V1530Ext;

    var IntegerList_Type v_RemovedDRBList := {};

    var integer v_NoOfDRBs := f_NR5GC_MobileInfo_GetNoOfDRBs();

    var NR_DRBInfoList_Type v_ExistingDRBs;

    var NR_DRBInfoList_Type v_NewDRBInfoList;

    var integer i;

    if (ispresent (p_RadioBearerConfig.drb_ToAddModList)) {

      v_NewDRBInfoList := fl_BuildDRBNewDRBList (valueof(p_RadioBearerConfig.drb_ToAddModList), valueof(p_SS_Config), valueof(p_CellGroupConfig)); //  @sic R5-196745 sic@

    }

    if (v_NoOfDRBs > 0) {

      v_ExistingDRBs := f_NR5GC_MobileInfo_GetDRBInfoList();

      v_ExistingDRBs := f_NR_DRBInfo_AddListOfEntries (v_ExistingDRBs, v_NewDRBInfoList);

    } else {

      v_ExistingDRBs := v_NewDRBInfoList;

    }

    // Remove DRBs about to be released

    if (ispresent (p_RadioBearerConfig.drb_ToReleaseList)) {

      v_RemovedDRBList := valueof (p_RadioBearerConfig.drb_ToReleaseList);

      for (i:= 0; i < lengthof(v_RemovedDRBList); i := i+1 ) {

        v_ExistingDRBs := f_NR_DRBInfo_RemoveEntry (v_ExistingDRBs, v_RemovedDRBList[i]);

      }

    }

    // Store updated DRB information

    f_NR5GC_MobileInfo_SetDRBInfoList(v_ExistingDRBs);

    // encode MasterCellGroup (just a list of RLC configs)

    v_V1530Ext := cs_NR_RRCReconfiguration_v1530_IEs(bit2oct(encvalue(p_CellGroupConfig)));

    //Configure DRB(s) in SS

    f_NR_SS_CommonRadioBearerConfig(p_CellId, p_SS_Config, p_TimingInfo);

    // enable ciphering for every new DRB

    v_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetInitial(p_CellId);

    f_NR_SS_RRC_EnableIntProt_CiphULandDL(p_CellId,

                                          v_SecurityParams.AS_Integrity,

                                          v_SecurityParams.AS_Ciphering);

    // Send message to UE

    f_NR_SendRRCReconfiguration (p_CellId, p_SRB, p_RadioBearerConfig, v_V1530Ext, p_NAS_Msg, -, p_TimingInfo);

  }
After Change
   function f_NR_RRCReconfigNewDRBWithSSConfig(NR_CellId_Type p_CellId,

                                              template (value) RadioBearerConfig p_RadioBearerConfig,

                                              template (value) NR_RadioBearerList_Type p_SS_Config,

                                              template (value) CellGroupConfig p_CellGroupConfig,

                                              template (omit) NG_NAS_MSG_RequestList_Type p_NAS_Msg,

                                              NR_RadioBearerId_Type p_SRB := tsc_NR_RbId_SRB1,

                                              template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now, 

                                              boolean p_ReceiveForRRCreconfigurationComplete := true) runs on NR5GC_PTC

  {

    var NR_SecurityParams_Type v_SecurityParams := f_NR_Security_Get();

    var template (value) RRCReconfiguration_v1530_IEs v_V1530Ext;

    var IntegerList_Type v_RemovedDRBList := {};

    var integer v_NoOfDRBs := f_NR5GC_MobileInfo_GetNoOfDRBs();

    var NR_DRBInfoList_Type v_ExistingDRBs;

    var NR_DRBInfoList_Type v_NewDRBInfoList;

    var integer i;

    if (ispresent (p_RadioBearerConfig.drb_ToAddModList)) {

      v_NewDRBInfoList := fl_BuildDRBNewDRBList (valueof(p_RadioBearerConfig.drb_ToAddModList), valueof(p_SS_Config), valueof(p_CellGroupConfig)); //  @sic R5-196745 sic@

    }

    if (v_NoOfDRBs > 0) {

      v_ExistingDRBs := f_NR5GC_MobileInfo_GetDRBInfoList();

      v_ExistingDRBs := f_NR_DRBInfo_AddListOfEntries (v_ExistingDRBs, v_NewDRBInfoList);

    } else {

      v_ExistingDRBs := v_NewDRBInfoList;

    }

    // Remove DRBs about to be released

    if (ispresent (p_RadioBearerConfig.drb_ToReleaseList)) {

      v_RemovedDRBList := valueof (p_RadioBearerConfig.drb_ToReleaseList);

      for (i:= 0; i < lengthof(v_RemovedDRBList); i := i+1 ) {

        v_ExistingDRBs := f_NR_DRBInfo_RemoveEntry (v_ExistingDRBs, v_RemovedDRBList[i]);

      }

    }

    // Store updated DRB information

    f_NR5GC_MobileInfo_SetDRBInfoList(v_ExistingDRBs);

    // encode MasterCellGroup (just a list of RLC configs)

    v_V1530Ext := cs_NR_RRCReconfiguration_v1530_IEs(bit2oct(encvalue(p_CellGroupConfig)));

    //Configure DRB(s) in SS

    f_NR_SS_CommonRadioBearerConfig(p_CellId, p_SS_Config, p_TimingInfo);

    // enable ciphering for every new DRB

    v_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetInitial(p_CellId);

    f_NR_SS_RRC_EnableIntProt_CiphULandDL(p_CellId,

                                          v_SecurityParams.AS_Integrity,

                                          v_SecurityParams.AS_Ciphering);

    // Send message to UE

    f_NR_SendRRCReconfiguration (p_CellId, p_SRB, p_RadioBearerConfig, v_V1530Ext, p_NAS_Msg, -, p_TimingInfo, p_ReceiveForRRCreconfigurationComplete);

  }
Change 4
	Function name
	f_NR5GC_CheckDataPath()

	Reason for change
	IP packets received from the UE maybe sent to SS IP stack unless the DRB routing table is explicitly updated to indicate raw mode handling

	Summary of change
	Change routing to raw mode before data path check

	TTCN module
	NR_CommonFunctions

	MCC160 Comment
	Rev1: Accepted in principle (see below).
In contrast to test loop mode, when using ping the UE is not in any kind of test mode and therefore e.g. IMS signalling may happen during the postamble ( the IP routing table needs to be restored.


Before Change
 function f_NR5GC_CheckDataPath(NR_CellId_Type p_CellId,

                              PDN_Index_Type p_PdnIndex,

                              DataPathCheck_BearerType_Type p_BearerType,

                              DRB_Identity p_DRBId,

                              charstring p_StepNo := "",

                              integer p_SequenceNumber := 1) runs on NR5GC_PTC

  {

   var octetstring v_SduTX := f_LoopbackModeBorPing_GetDataTX(p_PdnIndex, p_BearerType, p_SequenceNumber);

   var template (value) NR_DRB_COMMON_REQ v_DRB_DL;

   var template (present) NR_DRB_COMMON_IND v_DRB_UL;

   var template (value) QosFlow_Identification_Type v_QosFlowId;
   v_QosFlowId := f_NR_GetQosFlowIdentification_ForDRBId(p_DRBId);

   v_DRB_DL := cas_NR_DRB_COMMON_REQ_SDAP_SDUList(p_CellId, v_QosFlowId, { v_SduTX });

   v_DRB_UL := car_NR_DRB_COMMON_IND_SDAP_SDUList(p_CellId, v_QosFlowId, { v_SduTX });

   f_NR_CheckDataPathCommon(v_SduTX, v_DRB_DL, v_DRB_UL, p_StepNo);

 }
After Change

function f_NR5GC_CheckDataPath(NR_CellId_Type p_CellId,

                              PDN_Index_Type p_PdnIndex,

                              DataPathCheck_BearerType_Type p_BearerType,

                              DRB_Identity p_DRBId,

                              charstring p_StepNo := "",

                              integer p_SequenceNumber := 1) runs on NR5GC_PTC

  {

   var octetstring v_SduTX := f_LoopbackModeBorPing_GetDataTX(p_PdnIndex, p_BearerType, p_SequenceNumber);

   var template (value) NR_DRB_COMMON_REQ v_DRB_DL;

   var template (present) NR_DRB_COMMON_IND v_DRB_UL;

   var template (value) QosFlow_Identification_Type v_QosFlowId;
   f_IP_Routing_ChangeToRawmode(IP);
   v_QosFlowId := f_NR_GetQosFlowIdentification_ForDRBId(p_DRBId);

   v_DRB_DL := cas_NR_DRB_COMMON_REQ_SDAP_SDUList(p_CellId, v_QosFlowId, { v_SduTX });

   v_DRB_UL := car_NR_DRB_COMMON_IND_SDAP_SDUList(p_CellId, v_QosFlowId, { v_SduTX });

   f_NR_CheckDataPathCommon(v_SduTX, v_DRB_DL, v_DRB_UL, p_StepNo);

 }
MCC160 Implementation
	  type union IP_ConfigReq_Type {        /* @status    APPROVED (ENDC, IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, NBIOT, NR5GC, POS, UTRAN) */

    IP_StartPDN_Type                    StartPDN,

    IP_StopPDN_Type                     StopPDN,          /* @sic R5s131000: Null_Type replaced by IP_Stop_Type sic@ */

    IP_StartWLAN_Type                   StartWLAN,

    IP_StopWLAN_Type                    StopWLAN,

    Null_Type                           GeranCnf,         /* e.g. to confirm GeranTrigger */

    IP_ChangeDrbInfo_Type               ChangeDrbInfo,    /* @sic R5s150648 Change 5.1 sic@ */

    EUTRA_CellId_Type                   ChangeEutraCell,  /* @sic R5s150648 Change 5.1 sic@ */

    IP_SendRouterAdvertisement_Type     SendRouterAdvertisement,

    IP_ExplicitRouting_Type             ExplicitRouting,  /* @sic R5s130681 change 3: handling of RTP data; R5s160131: Start/Stop sic@ */

    Null_Type                           ChangeToRawmode,  /* @sic R5s150074 - Additional change sic@ */ 
    Null_Type                           RestoreRoutingTable,
    IP_DnsResponse_Type                 DnsResponse       /* @sic R5-169075 (test case 20.1) sic@ */

  };

	  template (value) IP_ConfigReq_Type cs_IP_RestoreRoutingTable := {RestoreRoutingTable := true};

	  function f_IP_Routing_RestoreRoutingTable(IP_RAT_CTRL_PORT p_Port)

  {

    fl_IP_Config(p_Port, cs_IP_RestoreRoutingTable);

  }

	  altstep a_ApplicationCtrl(IP_PTC_CTRL_PORT p_CtrlPort) runs on IP_PTC

  {

    var IP_ConfigReq_Type v_IP_ConfigReq;

    var PDN_Index_Type v_PdnIndex;

    var IP_DrbInfo_Type v_DrbInfo;

    var IP_DrbInfo_Type v_CurrentDrbInfo;

    var EUTRA_CellId_Type v_EutraCellId;

    var IP_RoutingTable_Type v_RoutingTable;
    var integer i;

    [] p_CtrlPort.receive(cr_IP_CtrlAnyReq) -> value v_IP_ConfigReq {

      if (ischosen(v_IP_ConfigReq.StartPDN)) {

        v_PdnIndex := v_IP_ConfigReq.StartPDN.PdnId;

        v_DrbInfo := v_IP_ConfigReq.StartPDN.DrbInfo;

        f_IP_PTC_PDN_SetDrbInfoForDefaultDrb(v_PdnIndex, v_DrbInfo);

        select (v_PdnIndex) {

          case (PDN_1, PDN_2, PDN_2a) {

            fl_IP_AddrServers_Start(v_PdnIndex, v_DrbInfo);

            fl_IP_ImsServer_Start(v_PdnIndex);

          }

          case (PDN_Internet, nrPDN_Internet, PDN_3) { // @sic R5s191074 sic@

            fl_IP_AddrServers_Start(v_PdnIndex, v_DrbInfo);

          }

          case (ePDG_IMS1, ePDG_IMS2) {

            fl_IP_ImsServer_Start(v_PdnIndex);

          }

          case (ePDG_XCAP) {

            fl_IP_AddrServers_Start(v_PdnIndex, v_DrbInfo, dnsServerOnly);

          }

          case (ePDG_Internet) {

            // nothing to be done

          }

        }

      } else if (ischosen(v_IP_ConfigReq.StopPDN)) {

        if (ischosen(v_IP_ConfigReq.StopPDN.PdnId)) {       /* @sic R5s131000: stop particular PDN sic@ */

          i := f_PDN_PdnIndex2Integer(v_IP_ConfigReq.StopPDN.PdnId);

          fl_StopPDN(i);

        } else {

          for (i:=0; i<tsc_NoOfPDNs; i:=i+1) {

            fl_StopPDN(i);

          }

        }

      } else if (ischosen(v_IP_ConfigReq.StartWLAN)) {

        fl_WLAN_Start(v_IP_ConfigReq.StartWLAN.AddressInfo, v_IP_ConfigReq.StartWLAN.DrbInfo, v_IP_ConfigReq.StartWLAN.ExplicitDnsHandling);

      } else if (ischosen(v_IP_ConfigReq.StopWLAN)) {

        fl_WLAN_Stop();

      } else if (ischosen(v_IP_ConfigReq.ChangeEutraCell)) {   /* @sic R5s150648 Change 6 sic@ */

        v_EutraCellId := v_IP_ConfigReq.ChangeEutraCell;

        if (f_RoutingTable_ChangeEutraCell(vc_IP_PTC_DRB_RoutingTable, v_EutraCellId)) {

          // we only change the routing at the SS when really needed

          f_IP_PTC_SendRoutingTable(vc_IP_PTC_DRB_RoutingTable);

          for (i:=0; i<tsc_NoOfPDNs; i:=i+1) {

            f_IP_DrbInfo_ChangeEutraCell(vc_IP_PTC_PDN_MappingTable[i].DefaultDrb, v_EutraCellId);

          }

        }

      } else if (ischosen(v_IP_ConfigReq.ChangeDrbInfo)) {   /* @sic R5s150648 Change 6 sic@ */

        v_PdnIndex := v_IP_ConfigReq.ChangeDrbInfo.PdnId;

        v_DrbInfo := v_IP_ConfigReq.ChangeDrbInfo.DrbInfo;

        v_CurrentDrbInfo := f_IP_PTC_PDN_GetDrbInfoForDefaultDrb(v_PdnIndex);

        if (f_RoutingTable_ChangeDrbInfo(vc_IP_PTC_DRB_RoutingTable, v_CurrentDrbInfo, v_DrbInfo)) {

          // we only change the routing at the SS when really needed

          f_IP_PTC_SendRoutingTable(vc_IP_PTC_DRB_RoutingTable);

          f_IP_PTC_PDN_SetDrbInfoForDefaultDrb(v_PdnIndex, v_DrbInfo);

        }

      } else if (ischosen(v_IP_ConfigReq.SendRouterAdvertisement)) {

        fl_ICMPv6_SendRouterAdvertisement(v_IP_ConfigReq.SendRouterAdvertisement.PdnId);

      } else if (ischosen(v_IP_ConfigReq.ExplicitRouting)) {

        if (ischosen(v_IP_ConfigReq.ExplicitRouting.Add)) {

          f_IP_PTC_RoutingTable_AddEntry(v_IP_ConfigReq.ExplicitRouting.Add.IpInfo,

                                         v_IP_ConfigReq.ExplicitRouting.Add.DRB,

                                         v_IP_ConfigReq.ExplicitRouting.Add.DataMode);

        } else if (ischosen(v_IP_ConfigReq.ExplicitRouting.Remove)) {  /* @sic R5s160131 sic@ */

          f_IP_PTC_RoutingTable_RemoveEntry(v_IP_ConfigReq.ExplicitRouting.Remove);

        }

      } else if (ischosen(v_IP_ConfigReq.ChangeToRawmode)) {  /* @sic R5s150074 - Additional change sic@ */


        v_RoutingTable := {}; 
        f_IP_PTC_SendRoutingTable(v_RoutingTable );

      } else if (ischosen(v_IP_ConfigReq.RestoreRoutingTable)) {
        v_RoutingTable := vc_IP_PTC_DRB_RoutingTable;
        f_IP_PTC_SendRoutingTable(v_RoutingTable);
      }

      p_CtrlPort.send(cs_IP_CommonCnf);

    }

  }

	  function f_NR_CheckDataPathCommon(octetstring p_SduTX,

                              template (value) NR_DRB_COMMON_REQ p_DRB_DL,

                              template (present) NR_DRB_COMMON_IND p_DRB_UL,

                              charstring p_StepNo := "") runs on NR_BASE_PTC

  { /* implementation as according to @sic Rs190936 sic@ */

   var NR_DRB_COMMON_IND v_NR_DRB_COMMON_IND;

   var octetstring v_SduRX;

   if (pc_IP_Ping) { f_IP_Routing_ChangeToRawmode(IP); }
   DRB.send(p_DRB_DL);

   DRB.receive(p_DRB_UL) -> value v_NR_DRB_COMMON_IND;

   if (pc_IP_Ping) { f_IP_Routing_RestoreRoutingTable(IP); }
   if (f_NR_Config_Get() == EN_DC) {

     v_SduRX := v_NR_DRB_COMMON_IND.U_Plane.SlotData.PduSduList.PdcpSdu[0];

   } else {

     v_SduRX := v_NR_DRB_COMMON_IND.U_Plane.SlotData.PduSduList.SdapSdu[0];

   }

   if (not f_LoopbackModeBorPing_CheckDataRX(p_SduTX, v_SduRX)) {

     f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid Echo Reply");

   } else {

     f_NR_PreliminaryPass(__FILE__, __LINE__, p_StepNo);

   }

 }

	  function f_EUTRA_CheckDataPath(EUTRA_CellId_Type p_CellId,

                                 PDCP_Protocol_Type p_PDCP_Protocol,

                                 DataPathCheck_BearerType_Type p_BearerType,

                                 charstring p_StepNo,

                                 template (value) RadioBearerId_Type p_DRB,

                                 integer p_SequenceNumber := 1,

                                 boolean p_UsePING := true) runs on EUTRA_PTC

  {

   var PDN_Index_Type v_PdnIndex := PDN_2;

   var octetstring v_PdcpSduTX := f_LoopbackModeBorPing_GetDataTX(v_PdnIndex, p_BearerType, p_SequenceNumber, p_UsePING);

   var PDCP_SDUList_Type v_PDCP_SDUList := { v_PdcpSduTX };

   var template (value) L2DataList_Type v_PDCP_DataTX;

   var template (present) L2DataList_Type v_PDCP_DataRX;

   var DRB_COMMON_IND v_DRB_COMMON_IND;

   var octetstring v_PdcpSduRX;

   var template (value) RlcBearerRouting_Type v_RlcBearerRouting := crs_RlcBearerRouting_Eutra(p_CellId);

   select (p_PDCP_Protocol) {

     case (PDCP) {

       v_PDCP_DataTX := cs_L2DataList_PdcpSdu(v_PDCP_SDUList);

       v_PDCP_DataRX := cr_L2DataList_PdcpSdu;

     }

     case (NR_PDCP) {

       v_PDCP_DataTX := cs_L2DataList_NrPdcpSdu(v_PDCP_SDUList);

       v_PDCP_DataRX := cr_L2DataList_NrPdcpSdu;

     }

   }

   if (pc_IP_Ping) { f_IP_Routing_ChangeToRawmode(IP); }
   DRB.send(cas_DRB_COMMON_REQ(p_CellId, p_DRB, cs_TimingInfo_Now, v_PDCP_DataTX, -, -, v_RlcBearerRouting));

   DRB.receive(car_DRB_COMMON_IND(p_CellId, p_DRB, cr_TimingInfo_Any, v_PDCP_DataRX, v_RlcBearerRouting)) -> value v_DRB_COMMON_IND;

   if (pc_IP_Ping) { f_IP_Routing_RestoreRoutingTable(IP); }
   select (p_PDCP_Protocol) {

     case (PDCP)    { v_PdcpSduRX := v_DRB_COMMON_IND.U_Plane.SubframeData.PduSduList.PdcpSdu[0]; }

     case (NR_PDCP) { v_PdcpSduRX := v_DRB_COMMON_IND.U_Plane.SubframeData.PduSduList.NrPdcpSdu[0]; }

   }

   if (not f_LoopbackModeBorPing_CheckDataRX(v_PdcpSduTX, v_PdcpSduRX)) {

     f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid Echo Reply");

   } else {

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, p_StepNo);

   }

  }


